Programming Language Semantics

Maude

Grigore Roșu
Department of Computer Science
University of Illinois at Urbana-Champaign
Maude is an *executable specification language* supporting *rewrite logic specifications* and having at its core a very fast *rewrite engine*.

We will use Maude as a platform for executing programming language definitions following various definitional styles. Our short term objectives are:

1. Learn how to use Maude;

2. Show how various language definitional styles, including the ones discussed so far or to be discussed (big-step and small-step SOS, MSOS, reduction semantics with evaluation contexts, the CHAM, denotational semantics) can be formally defined in Maude;
Maude’s roots go back to Clear (Edinburgh) in 1970s and to OBJ (Stanford, Oxford) in 1980s. There are many other languages in the same family, such as CafeOBJ (Japan), BOBJ (San Diego, California), ELAN (France).

Maude is currently being developed at Stanford Research Institute and at the University of Illinois at Urbana-Champaign. You can find more about Maude, including a manual and many papers and examples, on its webpage at http://maude.cs.uiuc.edu.

Due to the fact that Maude is executable, *interpreters for programming languages defined in Maude will be obtained for free*, which will be very useful for understanding, refining and/or changing the languages.
Moreover, *analysis tools* for the specified languages can also be obtained with little effort, such as type checkers, abstract interpreters or model-checkers, either by using analysis tools already provided by Maude or by deriving them explicitly from a language definition.

Devising *formal semantic executable models* of desired languages or tools *before these are implemented* is a crucial step towards a deep understanding of the language or tool.

In simplistic terms, it is like devising a *simulator* for an expensive system before building the actual system.
However, our “simulators” in this class will consist of exactly the *semantics*, or the *meaning*, of our desired systems, defined using a very rigorous, mathematical notation.

In the obtained formal executable model of a programming language, “executing a program” will correspond to nothing but *logical inference* within the semantics of the language.

**Maude** is easy to install and should run on the assigned machines. Type `maude` and you should see a welcome screen like the following:
bash$ maude

|
/|
--- Welcome to Maude ---
|
\|

Maude 2.5 built: May 7 2010 18:28:32
Copyright 1997-2010 SRI International
Wed Sep  1 10:16:41 2010

Maude>

You can download Maude from its web page at http://maude.cs.uiuc.edu. Unix/Linux is preferred, but it also works on Windows. Also save its user manual. You will need it.
How to Use Maude

Maude is interpreted; just type your specifications and commands.

Better way to use it:

1. Type everything in one file, say \texttt{p.maude}
2. Start Maude
3. Include that file with the command “\texttt{in p}” or “\texttt{in p.maude}”
4. Use “\texttt{quit}” or “\texttt{q}”, to quit.
Specifications are introduced as *modules*, or *theories*. We will use only general modules in this class, which have the syntax

```plaintext
mod <NAME> is
   <BODY>
endm
```

The `<BODY>` can include

- Importation of other modules
- Sort and operation declarations, forming the *signature* of that module; this is the interface of that module to the outside world
- Sentences.
We will use modules to define *programming language features*. These features can be then imported by other modules, which can extend them or use them to define other features. A programming language will be also defined as a module, namely one importing all the modules defining its desired features.

Let us exemplify this paradigm by defining the simplest programming language that we will see in this class. It is so simple, that it does not even deserve to be called a “programming language”, but we do so for the sake of a uniform terminology.

The language specified in what follows defines *natural numbers with addition and multiplication*. One module will define the feature “addition” and another will define the feature “multiplication” on top of addition.
An Example: Peano Natural Numbers

The following defines natural numbers with successor and addition, using Peano’s axiomatization:

```
mod PEANO-NAT is
  sort Nat .
  op zero : -> Nat .
  op succ : Nat -> Nat .
  op plus : Nat Nat -> Nat .
  vars N M : Nat .
  eq plus(zero, M) = M .
  eq plus(succ(N), M) = succ(plus(N, M)) .
endm
```

Declarations are separated by periods, which should typically have white spaces before and after.
Signatures

One sort, \texttt{Nat}, and three operations, \texttt{zero}, \texttt{succ}, and \texttt{plus}, form the signature of \texttt{PEANO-NAT}. Sorts are declared with the keywords \texttt{sort} or \texttt{sorts}, and operations with \texttt{op} or \texttt{ops}.

The three operations have zero, one and two arguments, resp., whose sorts are between : and \texttt{-}\. Operations of zero arguments are also called \textit{constants}, those of one argument are called \textit{unary} and those of two \textit{binary}. The result sort appears after \texttt{-}\. Use \texttt{ops} when two or more operations of same arguments are declared together, and use white spaces to separate them:

\begin{verbatim}
ops plus mult : Nat Nat -> Nat .
\end{verbatim}

There are few special characters in \texttt{Maude}. If you use \texttt{op} in the above then only one operation, called \texttt{``plus mult''}, is declared.
Equations

The two equations in **PEANO-NAT** are “properties”, or constraints, that these operations should satisfy. More precisely, any correct implementation of Peano natural numbers should satisfy them.

All equations are quantified universally, so we should read “for all $M$ and $N$ of sort $\textbf{Nat}$, $\text{plus}(s(N), M) = s(\text{plus}(N, M))$”. All equations in a module can be used in reasoning about the defined structures.

**Exercise 1** Prove that any correct implementation of **PEANO-NAT** should satisfy the property

$$\text{plus}(\text{succ}(\text{succ}(\text{zero})), \text{succ}(\text{succ}(\text{succ}(\text{zero})))) = \text{plus}(\text{succ}(\text{succ}(\text{succ}(\text{zero}))), \text{succ}(\text{zero})).$$

Equations can be applied from *left-to-right* or from *right-to-left* in reasoning, which means that equational proofs may require exponential search, thus making them theoretically intractable.
Maude regards all equations as *rewriting rules*, which means that they are applied only from left to right. Thus, any well-formed term can either be derived infinitely often, or be reduced to a *normal form*, which cannot be reduced anymore by applying equations as rewriting rules.

Maude’s command to reduce a term to its normal form is `reduce` or simply `red`. Reduction will be made in the last defined module:

```
Maude> reduce plus(succ(succ(zero)), succ(succ(succ(zero)))) .
rewrites: 3 in 0ms cpu (0ms real) (~ rewrites/second)
result Nat: succ(succ(succ(succ(succ(zero)))))
Maude>
```

*Millions of rewrites per second* can be performed, for which reason Maude can be (almost) used as a programming language.
Modules can be imported, using the keywords protecting, extending or including. The difference between these importation modes is subtle and semantical rather than operational. Feel free to use including, the most general of them, all the time and read the manual for more information.

The following module extends PEANO-NAT with multiplication:

```plaintext
mod PEANO-NAT* is
    including PEANO-NAT .
    op mult : Nat Nat -> Nat .
    vars M N : Nat .
    eq mult(zero, M) = zero .
    eq mult(succ(N), M) = plus(mult(N, M), M) .
endm
```

Variable declarations are not imported.
We can now define our simple programming language and “execute programs” (in this case, \texttt{PEANO-NAT\textdaggerdbl} already had all the features):

\begin{verbatim}
mod SIMPLEST-PROGRAMMING-LANGUAGE is
   including PEANO-NAT .
   including PEANO-NAT* .
endm
red mult(succ(succ(zero)), mult(succ(succ(zero)), succ(succ(zero)))) .
\end{verbatim}

The following is Maude’s output:

\begin{verbatim}
reduce in SIMPLEST-PROGRAMMING-LANGUAGE :
   mult(succ(succ(zero)), mult(succ(succ(zero)), succ(succ(zero)))) .
rewrites: 16 in 0ms cpu (0ms real) (~ rewrites/second)
result Nat: succ(succ(succ(succ(succ(succ(succ(succ(zero))))))))
\end{verbatim}

Even though this language is \textit{very simple} and its syntax is \textit{ugly}, it
nevertheless shows a formal and executable definition of a language using equational logic and rewriting.

Our languages or analysis tools discussed in this class will be defined in a relatively similar manner, though, as expected, they will be more involved. For example, a notion of state will be needed, as well as several control structures.
The Mix-fix Notation

Some operations are prefix, but others are infix, postfix, or mix-fix, i.e., arguments can occur anywhere, like in the case of `if_then_else_`.

Maude supports mix-fix notation by allowing the user to place the arguments of operations wherever one wants by using underscores. Here are some examples:

\begin{verbatim}
op _+_ : Int Int -> Int .
op !_ : Nat -> Nat .
op in_then_else_ : BoolExp Stmt Stmt -> Stmt .
\end{verbatim}

Exercise 2 Rewrite \texttt{PEANO-NAT} and \texttt{PEANO-NAT*} using mix-fix notation. What happens if you try to reduce an expression containing both \texttt{+_} and \texttt{*} without parentheses?
BNF (Bachus-Naur Form) or CFG (Context Free Grammar) notations to describe the syntax of programming languages are very common. We also used BNF to define the syntax of our simple language in the lectures on SOS. From here on, we can interchangeably use the mix-fix notation instead of BNF or CFG, so it is important to understand the relationship between the mix-fix notation and BNF and CFG:

**Exercise 3** Argue that the mix-fix notation is equivalent to BNF and CFG. Find an interesting simple example language and express its syntax using the mix-fix notation, BNF, and production-based CFG (e.g., $A \rightarrow AB \mid a$ and $B \rightarrow BA \mid b$).
Mix-fix notation leads to an interesting problem, that of parsing. In the modified PEANO-NAT* of Ex. 2, how do we parse \( x + y * z \)? 

Maude has a command called parse, which parses a term using the syntax of the most recently defined module. To see all the parentheses, first type the command “set print with parentheses on.” (see Subsection 17.5 in Maude’s manual).

“parse \( x + y * z \)” reports ambiguous parsing. Use parentheses to disambiguate parsing, such as “parse \( x + (y * z) \).”. To give priorities to certain operators in order to reduce the number of parentheses, use precedence attribute when you declare operators:

\[
\begin{align*}
\text{op } _+&\_ : \text{Int Int } \rightarrow \text{Int } [\text{prec 33}] . \\
\text{op } _*&\_ : \text{Int Int } \rightarrow \text{Int } [\text{prec 31}] .
\end{align*}
\]

The lower the precedence the stronger the binding!
Maude has several *builtin modules*, including ones for natural and integer numbers. The operations \_+_ and \_*_ have precedences 33 and 31, respectively, in these builtin modules. With these precedences, one can parse as follows:

Maude\> set print with parentheses on .
Maude\> parse -10 + 2 * 3 .

\text{Int: } (-10 + (2 * 3))

Sometimes, especially when debugging, the mixfix notation may be confusing. If you wish to turn it off, use the command:

Maude\> set print mixfix off .
Maude\> parse -10 + 2 * 3 .

\text{Int: } _+_(\text{-10, } _*_(2, \text{ 3}))
Many of the binary operations that will be used in this class will be associative (A), commutative (C) or have an identity (I), or combinations of these. E.g., \( + \) is associative, commutative and has 0 as identity. All these can be added as attributes to operations when declared:

\[
\text{op } _+_: \text{Int Int} \to \text{Int} \ [\text{assoc comm id: 0 prec 33}] . \\
\text{op } _*_: \text{Int Int} \to \text{Int} \ [\text{assoc comm id: 1 prec 31}] .
\]

Notice that each of the A, C, and I attributes are logically equivalent to appropriate equations, such as

\[
eq A + (B + C) = (A + B) + C .
\]

\[
eq A + B = B + A . \quad \text{---
 attention: rewriting does not terminate!}
\]

\[
eq A + 0 = A .
\]

Then why are the ACI attributes necessary? For several reasons:
1. Associativity removes the need for useless parentheses:

   Maude> parse -10 + 2 + 3 .
   Int: -10 + 2 + 3

2. Commutativity will allow rewriting to terminate. The normal form will, however, be *modulo commutativity*.

3. ACI matching is perhaps the most interesting reason. It will be described next.
Lists occur in many programming languages. The following module is a an elegant but tricky specification for lists of integers with a membership operation _in_:  

mod INT-LIST is protecting INT .  
    sort IntList .  
    subsort Int < IntList .  
    op nil : -> IntList .  
    op _in_ : Int IntList -> Bool .  
    var I : Int . vars L L’ : IntList .  
    eq I in L I L’ = true .  
    eq I in L = false [owise] .
endm
Note the declaration “\texttt{subsort Int < IntList .}”, which says that integers are also lists of integers. Then the constant \texttt{nil} and the concatenation operation \texttt{--} can generate any finite list of integers:

Maude> parse 1 2 3 4 5 .
IntList: 1 2 3 4 5
Maude> red 1 nil 2 nil 3 nil 4 nil 5 6 7 nil .
result IntList: 1 2 3 4 5 6 7

Further, by AI matching, the membership operation can be defined without having to traverse the list element by element!

Maude> red 3 in 2 3 4 .
result Bool: true
Maude> red 3 in 3 4 5 .
result Bool: true
Maude> red 3 in 1 2 4 .
result Bool: false

By AI matching, \texttt{I}, \texttt{L}, and \texttt{L'} in the left-hand-side term of “\texttt{eq I}
in L \ I \ L' = \texttt{true} .” can match any integer or lists of integers, respectively, including \texttt{nil}. Since \texttt{I} can only be matched by \texttt{3}, the other matches follow automatically.

The attribute \texttt{owise} in “eq I in L = false [owise] .” tells Maude to apply that equation only “otherwise”, that is, only if no other equation, in this case the previous one only, can be applied.

If one wants to define sets of integers, then, besides replacing the sort \texttt{IntList} probably by \texttt{IntSet}, one has to declare the concatenation also commutative, and one has to replace the first equation by “eq I in I L = true .”.

Exercise 4 Define a Maude module called \texttt{INT-SET} specifying sets of integers with membership, union, intersection and difference (elements in one set and not in the other).

The matching modulo attributes are implemented very efficiently in Maude, so you (typically) don’t have to worry about efficiency.
There are several built-in modules. We will use the following:

- **BOOL**. Provides a sort `Bool` with two constants `true`, `false` and basic boolean calculus, together with `if_then_else_fi` and `==` operations. The later takes two terms, reduces them to their normal forms, and then returns `true` iff they are equal (modulo ACI, as appropriate); otherwise it returns `false`.

- **INT**. Provides a sort `Int`, arbitrary large integers as constants of sort `Int`, together with the usual operations on these.

- **QID**. Provides a sort `Qid` together with arbitrary large quoted identifiers, as constants of sort `Qid`, of the form: `'a`, `'b`, `'a-larger-identifier`, etc.

To see an existing module, built-in or not, type the command `Maude> show module <NAME> .`
For example, “show module BOOL .” will output

mod BOOL is protecting TRUTH .
  op _or_ : Bool Bool -> Bool [assoc comm prec 59] .
  op _xor_ : Bool Bool -> Bool [assoc comm prec 57] .
  op _implies_ : Bool Bool -> Bool [prec 61 gather (e E)] .
  vars A B C : Bool .
  eq true and A = A .  eq false and A = false .
  eq A and A = A .  eq false xor A = A .
  eq A xor A = false .  eq not A = A xor true .
  eq A and (B xor C) = A and B xor A and C .
  eq A or B = A and B xor A xor B .
  eq A implies B = not (A xor A and B) .
endm

Exercise 5 Use the command “show module <NAME> .” on the three modules above and try to understand them.
When we defined the operation \texttt{plus} on Peano natural numbers, we defined it recursively on numbers of the form \texttt{zero} and \texttt{succ(N)}. Why did we do that, considering that \texttt{PEANO-NAT} had declared three operations:

\begin{verbatim}  
  op zero : -> Nat .  
  op succ : Nat -> Nat .  
  op plus : Nat Nat -> Nat .  
\end{verbatim}

Intuitively, the reason is that because we want \texttt{plus} to be completely \textit{defined} in terms of \texttt{zero} and \texttt{succ}. Formally, this means that any term over the syntax \texttt{zero}, \texttt{succ}, and \texttt{plus} can be shown, using the given equations, equal to a term containing only \texttt{zero} and \texttt{succ} operations, that is, \texttt{zero} and \texttt{succ} alone are sufficient to build any Peano number. For that reason, they are called \textit{constructors}.
Defining Operations using Constructors

There is no silver-bullet recipe on how to define “defined” operators, but essentially the main (safe) idea is to

*Define the operator’s “behavior” on each constructor.*

That is what we did when we defined \texttt{plus} in \texttt{PEANO-NAT} and \texttt{mult} in \texttt{PEANO-NAT*}: we first defined them on \texttt{zero} and then on \texttt{succ}.

In general, if \texttt{c1}, ..., \texttt{cn} are the intended constructors of a data-type, in order to define a new operation \texttt{d}, make sure that all equations

\[
\text{eq } d(c_1(X_1, \ldots)) = \ldots \\
\vdots \\
\text{eq } d(c_n(X_n, \ldots)) = \ldots
\]

are in the specification. This gives no guarantee (e.g., one can “define” \texttt{plus} as \texttt{plus(succ(M),N)} = \texttt{plus(succ(M),N)}), but it is a good enough principle to follow.
Defining Operations on Lists

Let us consider the following specification of lists:

\begin{verbatim}
mod INT-LIST is protecting INT .
    sort IntList . subsort Int < IntList .
    op nil : -> IntList .
endm
\end{verbatim}

Therefore, there are \textit{two constructors for lists}: the empty list and the concatenation of an integer to a list. Let us next define several other important and useful operations on lists. Notice that the definition of each operator treats each of the constructors separately.
The following defines the usual length operator:

mod LENGTH is protecting INT-LIST .

    op length : IntList -> Nat .
    var I : Int . var L : IntList .
    eq length(I L) = 1 + length(L) .
    eq length(nil) = 0 .
endm

red length(1 2 3 4 5) .  *** should be 5
The following defines membership, without speculating matching (in fact, this would not be possible anyway because concatenation is not defined associative as before):

```plaintext
mod IN is protecting INT-LIST.
  op _in_: Int IntList -> Bool.
  vars I J : Int. var L : IntList.
  eq I in J L = if I == J then true else I in L fi.
  eq I in nil = false.
endm

red 3 in 2 3 4.  ***> should be true
red 3 in 3 4 5.  ***> should be true
red 3 in 1 2 3.  ***> should be true
red 3 in 1 2 4.  ***> should be false
```
The next operator appends two lists of integers:

\[
\text{mod APPEND is protecting INT-LIST .}
\]

\[
\text{op append : IntList IntList -> IntList .}
\]

\[
\text{var I : Int . vars L1 L2 : IntList .}
\]

\[
\text{eq append(I L1, L2) = I append(L1, L2) .}
\]

\[
\text{eq append(nil, L2) = L2 .}
\]

\[
\text{endm}
\]

\[
\text{red append(1 2 3 4, 5 6 7 8) .}
\]

\[
***> \text{should be 1 2 3 4 5 6 7 8}
\]
The following imports **APPEND** and defines an operation which reverses a list:

mod REV is protecting APPEND .

    op rev : IntList -> IntList .
    var I : Int . var L : IntList .
    eq rev(I L) = append(rev(L), I) .
    eq rev(nil) = nil .

endm

red rev(1 2 3 4 5) .  ***> should be 5 4 3 2 1
The next module defines an operation which sorts a list of integers by insertion sort:

mod ISORT is protecting INT-LIST .
    op isort : IntList -> IntList .
    vars I J : Int . var L : IntList .
    eq isort(I L) = insert(I, isort(L)) .
    eq isort(nil) = nil .
    op insert : Int IntList -> IntList .
    eq insert(I, J L) = if I > J then J insert(I,L) else I J L fi .
    eq insert(I, nil) = I .
endm

red isort(4 7 8 1 4 6 9 4 2 8 3 2 7 9) .
***> should be 1 2 2 3 4 4 4 6 7 7 8 8 9 9
Let us now consider a specification of binary trees, where a tree can be either empty or an integer with a left and a right subtree:

mod TREE is protecting INT .
  sort Tree .
  op ___ : Tree Int Tree -> Tree .
  op empty : -> Tree .
endm

We next define some operations on such trees, also following the structure of the trees given by the two constructors above.
The next simple operation simply mirrors a tree, that is, it recursively replaces each left subtree by the mirrored right subtree and vice-versa:

mod MIRROR is protecting TREE .
    op mirror : Tree -> Tree .
    vars L R : Tree . var I : Int .
    eq mirror(L I R) = mirror(R) I mirror(L) .
    eq mirror(empty) = empty .
endm

red mirror((empty 3 (empty 1 empty)) 5 ((empty 6 empty) 2 empty)) .
***> should be (empty 2 (empty 6 empty)) 5 ((empty 1 empty) 3 empty)
Searching in binary trees can be defined as follows:

mod SEARCH is protecting TREE.

    op search : Int Tree -> Bool.
    vars I J : Int. vars L R : Tree.
    eq search(I, L I R) = true.
    eq search(I, L J R) = search(I, L) or search(I, R) [owise].
    eq search(I, empty) = false.
endm

red search(6, (empty 3 (empty 1 empty)) 5 ((empty 6 empty) 2 empty)).
***> should be true
red search(7, (empty 3 (empty 1 empty)) 5 ((empty 6 empty) 2 empty)).
***> should be false

Notice the use of the attribute \[owise\] for the second equation.

Exercise 6 Define \texttt{search} with only two equations, by using the built-in \texttt{if \_then\_else\_fi}. 
We next define a module which imports both modules of trees and of lists on integers, and defines an operation which takes a tree and returns the list of all integers in that tree, in an infix traversal:

mod FLATTEN is
    protecting APPEND .
    protecting TREE .
    op flatten : Tree -> IntList .
    vars L R : Tree . var I : Int .
    eq flatten(L I R) = append(flatten(L), I flatten(R)) .
    eq flatten(empty) = nil .
endm

red flatten((empty 3 (empty 1 empty)) 5 ((empty 6 empty) 2 empty)) .
***> should be 3 1 5 6 2

Exercise 7  Do the same for prefix and for postfix traversals.
Exercise 8 Write an executable specification in Maude that uses binary trees to sort lists of integers. You should define an operation \texttt{btsort : IntList -> IntList}, which sorts the argument list of integers, as \texttt{isort} did above. In order to define it, define another operation \texttt{bt-insert : IntList Tree -> Tree}, which inserts each integer in the list at its place in the tree, and also use the already defined \texttt{flatten} operation.